[Asus ROG Phone 7 / 7 Ultimate] cloneeye:
WW-AI2205-33.0820.0810.175-REL-2304-M2.6.16.15-ASUS_1.1.107-Phone-user-20230519.raw

Hmm... okay.

Then I will share it on my personal drive.

I found that this version worked flawlessly and had no problems.

@Echo off

REM Version: V10

REM Date: 20210824

REM ============== USAGE ==============

REM flashall_AFT.cmd [Format option] [SN number]

REM [Format option]: 1=enable, 0=disable. default=0

REM Format userdata/cache

REM [SN number]: device SN number

REM Support multiple download image

REM ============== setting ==============

SETLOCAL

SET LOG=flashraw_%2.log

IF EXIST %LOG% ( del %LOG% )

For /f %%i in ('date /t') do set mydate=%%i

For /f "tokens=1-2 delims=/:" %%a in ("%TIME%") do (set mytime=%%a:%%b)

ECHO TIMESTAMP : %mydate%_%mytime% >>%LOG%

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ########################### OS INFO ############################ >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

ECHO OS_VERSION = %PROCESSOR_ARCHITECTURE% >>%LOG%

IF "%PROCESSOR_ARCHITECTURE%" == x86 (

ECHO We don't support 32bit os environment, exit!

GOTO :EOF

)

IF "%selfWrapped%"=="" (

REM this is necessary so that we can use "exit" to terminate the batch file,

REM and all subroutines, but not the original cmd.exe

SET selfWrapped=true

%ComSpec% /s /c ""%~0" %*"

GOTO :EOF

)

REM ============== main() ==============

CLS

SET "dataerase=0"

SET "SSN="

SET "para="

SET "tempSysName=.system_tmp.img"

IF [%1] EQU [1] ( SET dataerase=1 )

IF NOT [%2] EQU [] ( SET para=-s )

IF NOT [%2] EQU [] (

SET SSN=%2

SET tempSysName=.system_tmp_%2.img

)

call:flash

EXIT 0

REM ============== flash() ==============

:flash

call:fastboot_tool flash all uboot 2>>%LOG%

call:fastboot_tool reboot-bootloader 1>>%LOG% 2>&1

ping 127.0.0.1 -n 6 > nul

echo "Start downloading AI2205 raw..."

FOR /F "delims=" %%i IN ('dir "%~dp0" /B ^| findstr /I ".raw$"') DO ( SET RAW=%%i )

IF NOT EXIST "%RAW%" ECHO "FAILED, RAW NOT EXIST!" && EXIT 1

ECHO.

ECHO ##################################################################

ECHO ##################### FLASH RAW FILE #####################

ECHO ##################################################################

ECHO.

ECHO ## RAW: %RAW%

ECHO ## ERASE DATA: %dataerase%

ECHO ## SSN: %SSN%

ECHO ## TIMESTAMP : %mydate%_%mytime%

ECHO.

ECHO ==================================================================

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ##################### FLASH RAW FILE ##################### >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

ECHO ## RAW: %RAW% >>%LOG%

ECHO ## ERASE DATA: %dataerase% >>%LOG%

ECHO ## SSN: %SSN% >>%LOG%

ECHO ## TIMESTAMP : %mydate%_%mytime% >>%LOG%

ECHO. >>%LOG%

ECHO ================================================================== >>%LOG%

fastboot version 1>>%LOG% 2>&1

ECHO ================================================================== >>%LOG%

call:fastboot_tool flash all "%RAW%" 2>>%LOG%

IF EXIST "APD.img" (

call:fastboot_tool flash APD APD.img 1>>%LOG% 2>&1

@ECHO Flashing APD successfully

)

REM call:fastboot_tool reboot-bootloader 2>>%LOG%

ping 127.0.0.1 -n 10 > nul

IF %dataerase% EQU 1 (

ECHO.

ECHO ##################################################################

ECHO ########################## WIPE DATA ###########################

ECHO ##################################################################

ECHO.

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ########################## WIPE DATA ########################### >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

call:fastboot_tool -w 1>>%LOG% 2>&1

call:fastboot_tool format asdf 1>>%LOG% 2>&1

call:fastboot_tool format ADF 1>>%LOG% 2>&1

call:fastboot_tool format logbuf 1>>%LOG% 2>&1

call:fastboot_tool erase asuskey5 1>>%LOG% 2>&1

call:fastboot_tool erase spunvm 1>>%LOG% 2>&1

call:fastboot_tool erase misc 1>>%LOG% 2>&1

)

set "ERRORLEVEL="

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ######################### DEVICE INFO ########################## >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

call:fastboot_tool oem device-info 1>>%LOG% 2>&1

call:fastboot_tool oem device-info 2>lock.txt && findstr /C:"(bootloader) Device unlocked: true" lock.txt

echo ERRORLEVEL=%ERRORLEVEL% 1>>%LOG% 2>&1

REM errorlevel == 0 => Device unlocked: true

IF %ERRORLEVEL% == 0 ( goto lock )

REM errorlevel == 1 => Device unlocked: false

IF %ERRORLEVEL% == 1 ( goto reboot )

REM ============== Lock ==================

:lock

IF %dataerase% EQU 1 (

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ######################### DEVICE LOCK ########################## >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

echo fastboot.exe oem asus-csc_lk >>%LOG%

call:fastboot_tool oem asus-csc_lk 1>>%LOG% 2>&1

)

REM ============== Reboot ================

:reboot

set "ERRORLEVEL="

findstr /C:"(bootloader) Device authorized: true" lock.txt

echo ERRORLEVEL=%ERRORLEVEL% 1>>%LOG% 2>&1

REM errorlevel == 0 => Device authorized: true

IF %ERRORLEVEL% == 0 ( goto oemlock )

REM errorlevel == 1 => Device authorized: false

IF %ERRORLEVEL% == 1 ( goto adb_enable )

REM ============== oem lock ================

:oemlock

IF %dataerase% EQU 1 (

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ####################### DEVICE OEM LOCK ######################## >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

echo fastboot.exe oem reset-dev_info >>%LOG%

call:fastboot_tool oem reset-dev_info 1>>%LOG% 2>&1

)

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ######################### DEVICE INFO ########################## >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

call:fastboot_tool reboot-bootloader 1>>%LOG% 2>&1

ping 127.0.0.1 -n 3 > nul

call:fastboot_tool oem device-info 1>>%LOG% 2>&1

ECHO ================================================================== >>%LOG%

call:fastboot_tool getvar partition-type:userdata 1>>%LOG% 2>&1

call:fastboot_tool getvar current-slot 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-retry-count:b 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-unbootable:b 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-successful:b 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-retry-count:a 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-unbootable:a 1>>%LOG% 2>&1

call:fastboot_tool getvar slot-successful:a 1>>%LOG% 2>&1

call:fastboot_tool getvar cid 1>>%LOG% 2>&1

REM ============= adb enable ===============

:adb_enable

ECHO. >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO ######################### ADB ENABLE ########################### >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

echo fastboot.exe oem adb_enable 1>>%LOG% 2>&1

call:fastboot_tool oem adb_enable 1>>%LOG% 2>&1

IF EXIST %tempSysName% DEL %tempSysName%

:DOWNLOAD_PASS

ECHO "Download_PASS"

ECHO.

ECHO ##################################################################

ECHO ################## FLASH COMPLETE. RESTART! ###################

ECHO ##################################################################

ECHO.

ECHO. >>%LOG%

For /f %%i in ('date /t') do set mydate=%%i

For /f "tokens=1-2 delims=/:" %%a in ("%TIME%") do (set mytime=%%a:%%b)

ECHO ## TIMESTAMP : %mydate%_%mytime% >>%LOG%

ECHO.

ECHO ################################################################## >>%LOG%

ECHO ################## FLASH COMPLETE. RESTART! ################### >>%LOG%

ECHO ################################################################## >>%LOG%

ECHO. >>%LOG%

GOTO:EOF

REM ============== fastboot_tool() ==============

:fastboot_tool

ECHO [command] : fastboot.exe %~1 %~2 %~3 %~4 %para% %SSN%

fastboot.exe %~1 %~2 %~3 %~4 %para% %SSN%

IF NOT %ERRORLEVEL% == 0 (

ECHO "FAILED, fastboot.exe %~1 %~2 %~3 %~4 %para% %SSN% failure, EXIT!"

IF EXIST %tempSysName% DEL %tempSysName%

EXIT 1

)

ECHO ==================================================================

GOTO:EOF

:END

@pause